
I’m Harjit Suman and I love all things Excel. I wrote this book as I want to help you advance your knowledge in Excel and for you to become as efficient as possible with your spreadsheets. One of the best ways to do this is to master Excel formulas. This book will teach you all the Excel functions and formulas you need to know to take your Excel skills to the next level. You will learn about all the commonly used Excel functions and formulas and then learn how to create more powerful formulas such as array formulas.

My Background

The first time I used Excel was in my first analytical role as a Business Analyst back in 2008. The problem was, I had no great experience in Excel other than simple data entry. Some of my first projects in data analysis took so long to do. In fact most of it was manually done as I knew of no other way.

I knew this couldn’t continue so I invested in a beginners Excel book to expand my knowledge. I wanted to automate my spreadsheets as much as I could to save me time. After reading my first book I started to understand the power of Excel. I bought more and more Excel books and practiced what I read in the evenings and weekends, well, whenever I had free time. I then applied what I learnt in my full time work as a Business Analyst. Soon, over time I went from an Excel beginner to a more advanced user and I managed to automate my spreadsheets as much as I could.

However, I wanted to take my Excel skills to another level so I learnt Visual Basic for Applications (VBA) where I could write my own macros. I bought VBA books and practiced what I learnt in my spare time. Now I am able to write my own macros to build tools and applications not just for me but for other businesses too.

Now it’s your Turn

Over the years I have learnt a lot about Excel and I now want to pass on my knowledge to you through this book. I have also created a website, www.excelmasterconsultant.com
. In here you can find reviews of the many Excel books I have read which have taken my Excel skills to the next level, and they will for you too. You will also find great tutorials and blogs I have written as well as Excel courses and add-ins you can buy. I also run an Excel consultancy service if you need any help with your spreadsheets.

I hope you enjoy this book. I would love to hear from you with anything Excel related so please get in touch by contacting me through my website.




About This Book

Who is this Book Aimed For?

This book is aimed for people who:


	Want to learn the true power of Excel formulas

	Already know how to create formulas but want to build more advanced formulas

	Know about some of the functions available in Excel but would like to learn about some of the more advanced Excel functions

	Want to advance their Excel skills from beginner or intermediate level to an advanced or expert level

	Have heard of array formulas but don’t know what they are and would like to know the true power of them



How this Book is Structured

I will first give you an overview of Excel and its history in the introduction. In the next chapter I will talk about the main Excel functions and formulas that are available which are used on regular basis by Excel users. I will also talk about the operator of order precedence in this chapter. Knowing this is critical in order to create formulas that give the correct results.

In Chapter two I will give you an introduction to arrays, array constants and array formulas. This will give you a good grounding and understanding for the chapters that follow as they contain more advanced formulas such as array formulas.

For Chapters 3 to 12 you will learn some awesome formulas! Each of these chapters contains a business case scenario and the objective of what the goal is. I then list the Excel functions that are needed in the formula to achieve the goal. Next I explain in more detail what each function does, what the syntax is and a description of its arguments. I then show you the formula which will solve the problem. Finally I will explain how the formula works by breaking it down step by step.

In the last chapter I will give you some golden nuggets of information, especially about array formulas.

Tip

The formulas I mention in chapters 3 to 12 are complex. I have explained how they work by breaking the formula down. To truly understand what each section of the formula does you can highlight a functions argument and press F9 on your keyboard to display its output.

For example:


	Understanding how the below formula calculates is difficult when it is not broken down in steps



=MID(B3,SEARCH("-",B3)+1,SEARCH("-",B3,SEARCH("-",B3)+1)-SEARCH("-",B3)-1)


	If I highlight a MID argument like this:



=MID(B3,SEARCH("-",B3)+1
,SEARCH("-",B3,SEARCH("-",B3)+1)-SEARCH("-",B3)-1)


	Then press F9 on my keyboard it will display the output for that argument like this:



=MID(B3,5,SEARCH("-",B3,SEARCH("-",B3)+1)-SEARCH("-",B3)-1)


	The highlighted section in step 2 returns a 5




Introduction

For those of you who are interested, here is a brief history lesson of Excel and its background. We normally take Excel for granted as large and small organisations use it on a daily basis to store data, record financial information, create applications and models and much more but very few of us know its origins.

What is Еxcеl?

Microsoft Еxcеl is a sprеadshееt application that allows user’s tо stоrе, еdit and manipulatе data within a grоup оf table like structurеs knоwn as a wоrkbооk. A wоrkbооk usеs a cоllеctiоn оf wоrkshееts tо organize data within rоws and cоlumns split intо cеlls. Usеrs can еntеr data оr numbеrs intо thеsе cеlls tо crеatе еasy ways tо calculatе simplе оr cоmplеx fоrmulas. Usеrs can insеrt оbjеcts such as charts and graphs intо thеir wоrkbооk tо graphically rеprеsеnt thеir еntеrеd data in many uniquе ways.

Whо Crеatеd Еxcеl?

Microsoft Еxcеl was crеatеd by thе Micrоsоft Cоrpоratiоn in 1984. It was dеsignеd tо crеatе a mоrе еfficiеnt user-friendly way fоr sprеadshееt usеrs tо calculatе data withоut thе wоrry оf having tо dеal with thе DОS cоmmand linе. It was originally crеatеd fоr usе with Applе cоmputеrs. Whеn Micrоsоft Windоws was first launchеd in 1987, Micrоsоft Еxcеl was оnе оf thе first prоgrams crеatеd fоr Micrоsоft Windоws. Until 1992 it was thе оnly windоws basеd sprеadshееt applicatiоn.

Vеrsiоns

Micrоsоft Еxcеl crеatеd sеvеral diffеrеnt vеrsiоns fоr thrее diffеrеnt оpеrating systеms. Micrоsоft Windоws has had 11 diffеrеnt vеrsiоns frоm 1987 until nоw. Applе Macintоsh has had 12 diffеrеnt vеrsiоns frоm 1985 until nоw. Thе ОS/2 оpеrating systеm had 3 diffеrеnt vеrsiоns frоm 1989 tо 1991.

Filе Typеs

Micrоsоft Еxcеl has alsо crеatеd nеw typеs оf ways tо savе wоrkbооks and оthеr variоus Еxcеl filеs. With thе crеatiоn оf nеw typеs оf saving filеs, Еxcеl crеatеd thе impоrt and еxpоrt fеaturе that allоwеd Еxcеl usеrs tо еasily bе ablе tо impоrt alrеady crеatеd wоrkbооks оr filеs and еxpоrt wоrkbооks оr filеs to be usеd at a latеr timе.

List оf Filе Typеs

XLS – Еxcеl Filе

This еxtеnsiоn is thе mоst cоmmоn and dеfault typе in thе sprеadshееt gеnеratеd by Micrоsоft Officе. Priоr tо Еxcеl 2007, thе filе еxtеnsiоn was XLS. This filе rеfеrs tо a filе which cоntains all typеs оf infоrmatiоn including data, fоrmats, graphics, еtc.

XLSX – Еxcеl Filе

This filе is usеd by thе sprеadshееt filеs gеnеratеd with Еxcеl vеrsiоn 2007 оnwards. Fоr an Еxcеl filе, thе currеnt dеfault filе еxtеnsiоn is XLSX.

XLSM – Еxcеl Filе

This filе typе is gеnеratеd by thе sprеadshееt with Excеl vеrsiоn 2007 оnwards including Еxcеl macrоs. With thе hеlp оf this еxtеnsiоn, its еasy tо undеrstand that thе filе cоntains a macrо. Bеcausе оf sеcurity rеasоns and fоr prоtеcting a filе with cоmputеr virusеs, maliciоus macrоs, infеcting cоmputеrs, еtc. this vеrsiоn came intо еxistеncе.

XLSB – Еxcеl Filе

If Excеl filеs cоntain a largе amоunt оf data оr infоrmatiоn, this filе еxtеnsiоn typе fully suppоrts in thе cоmprеssiоn prоcеss, saving, оpеning, еtc. An Excеl filе which cоntains a largе amоunt оf data takеs a lоt оf timе in оpеning and prоcеssing that filе. Sоmеtimеs whilе оpеning it gеts hangеd and frеquеntly crashes.


Chapter 1: Excel Formulas and Functions

In this chapter I have listed some of the most useful Excel functions available with examples of how to use them. This is not a complete list of Excel functions but a list of my favourite functions that I use on a regular basis and that have not already been explained in the chapters later in this book.

Later in this chapter I will explain the order in which Excel performs its calculations. This is called operator precedence and is very important when building formulas, especially large more complex formulas.

Sum Functions

Count Functions

Logical Functions

Lookup Functions

Text Functions

Other Functions

Order of Operator Precedence

A very important component in building formulas is to understand the order in which Excel performs its calculations.

Below is a table which shows the order of operations in Excel:

At the very top is using parenthesis. You can force Excel to perform the calculation you want by using parenthesis to specify what operation to calculate first. Note the multiplication and division have the same order precedence as well as the comparison operators. If the formula contains operators with the same order precedence then Excel will evaluate the operators from left to right.

Let’s now have a look with an example of how Excel orders a calculation in a formula, with and without parenthesis.

If we take the formula =6+7*4, Excel will return the answer 34. This is because Excel will calculate 7*4 first as multiplication has a higher order of precedence than addition. Now if we use parenthesis like this =(6+7)*4, Excel will return the answer 52. 6+7 is calculated first as it has parenthesis around it. This returns the answer 13 and then 13 is multiplied by 4 to give the answer 52.

Now let’s look at a more complicated example where a formula has multiple parenthesis. When a formula has multiple parenthesis, Excel will calculate it from the inside out. Below is a step by step instruction of how the formula =5*(100-(4*4)/2) is worked out.


	First Excel calculates (4*4) and returns the answer 16. The formula calculation is =5*(100-16/2)




	Now Excel will calculate 16/2 next as division has a higher precedence than subtraction. The formula calculation is =5*(100-8)




	Now Excel will calculate (100-8) which returns the answer 92. It will then multiply this by 5. The formula calculation is =5*92 and the result is 460



Understanding the order of operator precedence is critically important in order to construct formulas, especially larger more complex formulas correctly. Using parenthesis in the correct way is critical in order to get the correct results.

A common problem with using parenthesis is that when you start to open them you forget to close them all. Excel will display an error message saying it found a mistake and will offer to close the formula with a parenthesis. The problem with this is that if you click the Yes button it will put the closing parenthesis at the end of the formula which sometimes is not correct and will therefore display an incorrect result. It is important to know where the closing parenthesis should be located.


Chapter 2: Introduction to Array Formulas

Many of the formulas I talk about in the following chapters are array formulas. If you want to master Excel to an advanced level and create elegant formulas that perform what looks like magic then you need to understand array formulas.

But first, what exactly is an array formula?

What is an Array Formula?

To understand what an array formula is you first need to understand what an array is. An array is a collection of items. It can be one dimensional or two dimensional and the dimensions can be in rows or columns. A one dimensional array can be stored in a single row or column. A two dimensional array can be stored in a range of cells of a rectangular shape. Arrays appear in curly brackets { } such as {“Jan”, “Feb”, “Mar”}. When items are in curly brackets it is called an array constant. I will explain more about array constants in the next section. An interesting fact about arrays is that they don’t have to be stored in cells. It can also be stored in memory.

An array formula will perform calculations on the arrays. One important thing to note about array formulas is that on most occasions you have to press CTRL+SHIFT+ENTER on your keyboard. By doing this it will enter curly brackets { } before and after the formula. This means that the formula is an array formula. Notice I said on most occasions you have to do this. Depending on what function you use you do not always have to press CTRL+SHIFT+ENTER. For example, if you are using the SUMPRODUCT function. You will see some examples later on where you don't have to press CTRL+SHIFT+ENTER for an array formula. I have listed which functions require you to press CTRL+SHIFT+ENTER and for which functions you don’t need to in Chapter 13.

An important point to make is that for the array formulas that require curly brackets you MUST press CTRL+SHIFT+ENTER and not just type them in manually from your keyboard. If you do then the formula will return an error.

What are Array Constants?

I mentioned earlier that arrays don’t have to be stored in cells but can also be stored in memory. This is referred to as an array constant. An array constant is when items are listed in curly brackets { }.

Here is an example of an array constant that has 6 items:

{1,1,0,4,6,0}

An array constant on its own is pretty useless. You start to see the benefit of them when you combine other functions with them such as a SUM function like this:

=SUM({1,1,0,4,6,0})

This SUM function simply adds each item in the array constant together to return the answer 12. Note this is not an array formula so you don’t need to press CTRL+SHIFT+ENTER.

You might think what is the point of this formula when you can just use =1+1+0+4+6+0 or if the values are in a range such as A1:A6 you can use =SUM(A1:A6). Well you will start to understand the power of array constants in my next example.

This is a formula that uses two array constants:

=SUM({2,4,6,8}*{10,12,14,16})

This formula multiples the first items from each array together then the second items together and so on. The formula calculation will look like this:

=SUM(2*10, 4*12, 6*14, 8*16)

This calculation creates an array which is stored in memory:

{20, 48, 84, 128}

This array in memory is passed to the SUM function to give the answer 280.

Array Formula Example

Now you know what array formulas and array constants are let’s look at an example of an array formula.

In this example there are two sales people with their sales in Jan to Apr. I want to extract the sales for both sales people in March.

The array formula used to extract the sales for March for Sales person 1 is:

=SUM(IF($A$2:$A$5="Mar",$B$2:$B$5,0))

The array formula used to extract the sales for March for Sales person 2 is:

=SUM(IF($A$2:$A$5="Mar",$C$2:$C$5,0))

So how does this array formula work? Let’s investigate further using the formula for Sales person 1.


	First the array formula is testing if the values in the range A2:A5 is “Mar”. The array will show:



{FALSE;FALSE;TRUE;FALSE}


	The second part of the array formula $B$2:$B$5 is listing the sales in the range B2:B5 like this:



{500;800;100;900}


	The formula calculation is this



=SUM(IF({FALSE;FALSE;TRUE;FALSE},{500;800;100;900},0))


	The formula will return £100 as this is the third item in the second array and in the first array the third item is TRUE.



Note you can perform the same calculation using the SUMIF function but this example will give you an idea of how array formulas work. Array formulas really come into their own when there is no other way to perform the task.

What are the Advantages of Array Formulas?

So what are the advantages of using array formulas?


	They can perform tasks which would normally be impossible to achieve using a regular formula




	They can replace intermediate formulas so can save on spreadsheet space




	They can make your work much more efficient




	Array formulas increase your ability to create more efficient formulas, enhance your creativity and improve your formula skills



What are the Disadvantages of Array Formulas?

Array formulas have many advantages but it’s also important to know that it also has some disadvantages. It is always best to assess if the task can be performed using a regular formula first. Some of the disadvantages are:


	Array formulas can be very slow so if you have a large data set the calculation will take a lot longer to complete




	Array formulas can be difficult to understand for beginners so if you share the workbook with them and the worksheet contains errors they will not be able to fix it




	
You must remember to press CTRL+SHIFT+ENTER otherwise the formula will give an error








Chapter 3: Find the First Non-Blank Value

Objective

Find the first sales figure for each sales person. For example, for Jim the first non-blank cell is in March so the formula needs to extract £5,400. For Peter the first non-blank cell is in February so the formula needs to extract £5,000 and so on.

Functions Used


	INDEX

	MATCH

	ISBLANK



INDEX Function Explained

The INDEX function retrieves a value from a data set. You have to specify the position of the value you want to retrieve by choosing a row number and optionally a column number in the data set. The INDEX function is often used with the MATCH function to create powerful lookup formulas.

Syntax

The syntax for the INDEX function is:

=INDEX(array, row_num, [column_num])

Arguments

array – A range of cells or an array constant

row_num – The row number in the array

column_num - This is an optional argument. This is the column number in the array

MATCH Function Explained

The MATCH function is used to get the position number of a lookup value in a row or column in the data set. You can use approximate and exact match to get the position of the lookup value in a data set. The MATCH function is often used with the INDEX function.

Syntax

The syntax for the MATCH function is:

=MATCH(lookup_value, lookup_array, [match_type])

Arguments

lookup_value – The value you want to match in the data set (lookup_array)

lookup_array – The range of cells where you want to find the lookup_value

match_type – This is an optional argument. 1 indicates a less than match, 0 indicates an exact match, -1 indicates a greater than match

ISBLANK Function Explained

The ISBLANK function returns TRUE if a cell contains no numbers or text, i.e. the cell is empty. It returns FALSE if a cell contains a number of text, i.e. the cell is not empty.

Syntax

=ISBLANK(value)

Arguments

value – the cell to check if it is empty or not

The Formula to Extract the First Non-Blank Value

In cell P5 the formula to extract the first non-blank value for Jim is:

=INDEX(C5:N5,1,MATCH(1,INDEX(1-ISBLANK(C5:N5),1,0),0))

The formula is then copied down to cell P9.

How the Formula Works


	The ISBLANK function will return TRUE if a cell is blank and FALSE if the cell is not blank




	Range C5:N5 will return TRUE, TRUE, FALSE, TRUE, FALSE, TRUE, TRUE, FALSE, TRUE, TRUE, TRUE, TRUE




	The TRUE and FALSE values are either a 1 or a 0 in an array formula. TRUE = 1 and FALSE = 0




	The 1-ISBLANK(C5:N5) will therefore return a 0 for a blank cell and a 1 for a non-blank cell. For example C5 will return a 0 because 1-1 = 0. Cell E5 will return a 1 because 1-0 = 1




	INDEX(1-ISBLANK(C5:N5),1,0) will therefore return the array {0,0,1,0,1,0,0,1,0,0,0,0}




	MATCH(1,INDEX(1-ISBLANK(C5:N5),1,0),0) returns the first position of 1 in the array {0,0,1,0,1,0,0,1,0,0,0,0}. In this array the first position of 1 is in position 3, i.e. March. If you break this formula down it will look like this MATCH(1, {0,0,1,0,1,0,0,1,0,0,0,0},0)




	You know that the third column contains the first non-blank cell. You now need to return the value in this third column. This is where you use the INDEX function. To break this down the formula will be =INDEX(C5:N5,1,3)




	The result is £5,400




	Note: if all the cells are blank then it will return a #N/A error




	There is an alternative formula which also finds the first non-blank cell. In cell Q5 the formula is:



=INDEX(C5:N5,1,MATCH(FALSE,INDEX(ISBLANK(C5:N5),1,0),0))

This formula is similar to the first method but this one looks for FALSE as the first argument of the MATCH function


Chapter 4: Extract a Unique List

Objective

Extract a list of unique values from a list of duplicate values.

In the range B3:B11 there is a list of first names. Some names are duplicates. In the range C3:C11 I want to extract just the unique names from B3:B11.

Functions Used


	IFERROR

	INDEX

	MATCH

	COUNTIF



IFERROR Function Explained

The IFERROR function is an error handler. It returns a text string or value you specify if the formula returns an error. It is a more efficient way to handle errors than nested IF statements.

Syntax

=IFERROR(value, value_if_error)

Arguments

value – The cell, value or formula to check for an error

value_if_error – The value to specify if there is an error

INDEX Function Explained

Please see Chapter 3 – Find the First Non-Blank Value

MATCH Function Explained

Please see Chapter 3 – Find the First Non-Blank Value

COUNTIF Function Explained

The COUNTIF function counts the number of cells based on a single criteria. You can use logical operators (>, <, <>, =) for the COUNTIF function to count the cells.

Syntax

=COUNTIF(range, criteria)

Arguments

range – The range of cells you want to count

criteria – The criteria to use to count the cells

The Formula to Extract the Unique List

In cell C3 the formula used to extract the unique list is:

{=IFERROR(INDEX($B$3:$B$11,MATCH(0,COUNTIF($C$2:C2,$B$3:$B$11),0)),"")}

The formula is then copied down to cell C11.

How the Formula Works


	I will start with the COUNTIF($C$2:C2,$B$3:$B$11) section of the formula first. The first argument of the COUNTIF function is the range $C$2:C2 which returns the string "Unique List". The second argument in the COUNTIF function checks if this string is in the range B3:B11. There isn't a "Unique List" value in this range so it returns an array of {0;0;0;0;0;0;0;0;0}




	MATCH(0,COUNTIF($C$2:C2,$B$3:$B$11),0) checks to see where the first position of the 0 is in the array {0;0;0;0;0;0;0;0;0}. We can see there is a 0 in the very first position so this returns a 1




	We need to return the first value from the range B3:B11 and we do this with an INDEX function. If we break this formula down it will become =INDEX($B$3:$B$11,1) which returns Sarah




	The IFERROR function is used to return any #N/A errors into empty strings



Points to Note


	The source data cannot contain any empty cells otherwise the formula will start to return zeros




	The range in the COUNTIF functions first argument starts in the row one above the formula. In this example it is cell C2




	You must press CTRL+SHIFT+ENTER otherwise the formula will not work




Chapter 5: Extract the Last Non-Empty Cell

Objective

Extract the last non-empty value in a list.

In the range B2:C12 there is a list of the last 10 premier league winners. I want to extract the last team that won the premier league.

Functions Used


	LOOKUP



LOOKUP Function Explained

The LOOKUP function is used to lookup a value from a single row or column and returns the value from another row or column in the same position as the lookup value.

Syntax

=LOOKUP(lookup_value, lookup_vector, [result_vector])

Arguments

lookup_value – The value to look for in a single row or column

lookup_vector – The single row or column to search

result_vector – This is an optional argument. It is the single row or column range of results

The Formula to Extract the Last Non-Empty Cell

In cell F2 the formula used to extract the last non-empty cell is:

=LOOKUP(2,1/(C:C<>""),C:C)

How the Formula Works


	The LOOKUP function looks at the value in the first argument (lookup_value) and returns a value from its third argument (result_vector)




	If we look at the LOOKUP functions second argument (lookup_vector) C:C<>"" it returns a series of TRUE and FALSE values




	The number 1 is then divided by this array of TRUE and FALSE values. If the value in column C is not blank it returns a 1 and if it is blank it will return a #DIV/0! error




	When the first argument, i.e. the number 2 cannot be found it will then look up the next smallest value, i.e. 1




	The LOOKUP function will therefore look at the last 1 value in the array




	The LOOKUP functions third argument will return the value of the last 1 value in the array




	The result is Manchester City



Points to Note


	The LOOKUP functions first argument (lookup_value) is deliberately larger than any values that will appear in the LOOKUP functions third argument (vector result)




Chapter 6: Sum all Digits in a String

Objective

Sum all digits in a text string.

In the above table there are text strings in the range B3:B11 with numbers in them. Normally if there are numbers in a text string they are not actual values so you can’t do any sums with them. Here I want to add the digits together even though they are in a text string.

Functions Used


	SUM

	LEN

	SUBSTITUTE



SUM Function Explained

The SUM function is one of the most commonly used functions in Excel. It simply adds the values in cells together to give a total. The SUM function can handle 255 arguments.

Syntax

=SUM(number1, [number2],…)

Arguments

number1 – The first value to sum

number2 – This is optional. The second value to sum

LEN Function Explained

The LEN function returns the length of a text string. The LEN function counts a space as a character so is included in the calculation.

Syntax

=LEN(text)

Argument

text – The text to use to calculate its length

SUBSTITUTE Function Explained

The aim of the SUBSTITUTE function is to replace existing text with a new text in a string. The SUBSTITUTE function is case sensitive. An alternative function to use is the REPLACE function.

Syntax

=SUBSTITUTE(text, old_text, new_text, [instance_num])

Arguments

text – The text to change

old_text – The text you want to replace

new_text – The new text to replace the old_text with

instance_num – This is optional. The instance to replace

The Formula to Sum all Digits in a String

In cell C3 the formula to add the digits from the text string in cell B3 is:

=SUM((LEN(B3)-LEN(SUBSTITUTE(B3,{1,2,3,4,5,6,7,8,9},"")))*{1,2,3,4,5,6,7,8,9})

The formula is then copied down to cell C11.

How the Formula Works


	We will look at the SUBSTITUTE(B3,{1,2,3,4,5,6,7,8,9},"") section of the formula in cell C3 first. The numbers 1 to 9 are substituted with an empty string. This returns 9 string values without any corresponding digits




	Because there are 2 occurrences of 4 in the string in cell B3, it returns 2 less than that of the original string in LEN(B3). The formula =LEN(B3)-LEN(SUBSTITUTE(B3,4,"")) will return 2




	The formula LEN(B3)-LEN(SUBSTITUTE(B3,{1,2,3,4,5,6,7,8,9},"")) returns an array of 9 values which will give you the number of occurrences of each digit in a string in an array, i.e. {0,1,0,2,0,0,0,1,0}. For example, there are no 1's in cell B3, there is a 2, no 3's, two 4's etc




	SUM((LEN(B3)-LEN(SUBSTITUTE(B3,{1,2,3,4,5,6,7,8,9},"")))*{1,2,3,4,5,6,7,8,9}) will therefore calculate as =SUM({0,1,0,2,0,0,0,1,0}*{1,2,3,4,5,6,7,8,9})




	The formula then sums the two products of the arrays e.g. 0x1 + 1x2 + 0x3 + 2x4 + 0x5 + 0x6 + 0x7 + 1x8 + 0x9




	The result is 18




	There is an alternative method where you can replace the SUM function with SUMPRODUCT and replace the multiplication sign with a comma. The formula is:



=SUMPRODUCT(LEN(B3)-LEN(SUBSTITUTE(B3,{1,2,3,4,5,6,7,8,9},"")),{1,2,3,4,5,6,7,8,9})


Chapter 7: Extract an Email Address from a Text String

Objective

Extract an email address from a text string.

Range B3:B5 contains text strings and within the text strings are some email addresses. The objective is to extract just the email addresses from the text strings.

Functions Used


	TRIM

	MID

	SUBSTITUTE

	REPT

	FIND



TRIM Function Explained

The TRIM function removes any spaces before and after a text. It also leaves a single space between words.

Syntax

=TRIM(text)

Argument

text – The text where spaces are removed

MID Function Explained

The MID function extracts characters from inside a text string. You have to specify the starting point and the number of characters you want to extract from inside the text string.

Syntax

=MID(text, start_num, num_chars)

Arguments

text – The text you want to use to extract the characters from the inside

start_num – The position in the text you want to start the extraction

num_chars – The number of characters you want to extract from the text

SUBSTITUTE Function Explained

See Chapter 6 – Sum all Digits in a String

REPT Function Explained

The REPT function repeats characters a given number of times. A little known feature of the REPT function is that you can create histograms. You can see my blog on how to do this:


https://www.excelmasterconsultant.com/single-post/2018/07/26/How-to-Create-a-Histogram-in-Excel


Syntax

=REPT(text, number_times)

Arguments

text – The text you want to repeat

number_times – The number of times you want to repeat the text

FIND Function Explained

The FIND function finds the position of a character in a text string. The FIND function is case sensitive.

Syntax

=FIND(find_text, within_text, [start_num])

Arguments

find_text – The character you want to find in the text string

within_text – The text string to search in

start_num – This is optional. The starting position in the text string to search

The Formula to Extract an Email Address from a Text String

The formula to extract an email address from a text string in cell C3 is:

=TRIM(MID(SUBSTITUTE(" "&B3," ",REPT(" ",20)),FIND("@",SUBSTITUTE(" "&B3," ",REPT(" ",20)))-20,50))

The formula is then copied down to cell C5.

How the Formula Works


	Let’s break the formula in cell C3 down and start with SUBSTITUTE(" "&B3," ",REPT(" ",20)). This section of the formula substitutes every occurrence of a space with 20 spaces using the REPT function. This returns "                    Write                    to                    Jim                    Smith                    at                    jsmith@yahoo.com                    for                    your                    Excel                    quote"




	The FIND function finds the @ in the new text in this section of the formula:



FIND("@",SUBSTITUTE(" "&B3," ",REPT(" ",20))) and returns 144


	MID(SUBSTITUTE(" "&B3," ",REPT(" ",20)),FIND("@",SUBSTITUTE(" "&B3," ",REPT(" ",20)))-20 means the MID function starts 20 characters before the @




	MID(SUBSTITUTE(" "&B3," ",REPT(" ",20)),FIND("@",SUBSTITUTE(" "&B3," ",REPT(" ",20)))-20,50) extracts the text for the next 50 characters. 50 is used as the MID Functions third argument (num_chars) as this will be long enough to extract a full email address. You can use any number as long as you know it will be long enough to extract the full email address




	The TRIM function is used to remove all the spaces before and after the email address




	The result is jsmith@yahoo.com




Chapter 8: Two Way Lookup

Objective

To perform a two way lookup to extract the sales figure for a Sales person for a particular month.

[image: ]


I want to extract the sales figure for Abida in June.

Functions Used


	INDEX

	MATCH



INDEX Function Explained

Please see Chapter 3 – Find the First Non-Blank Value

MATCH Function Explained

Please see Chapter 3 – Find the First Non-Blank Value

The Formula to Extract the Sales Figure for a Sales Person in a Given Month

The formula to extract the sales figure for Abida in June is:

=INDEX(C3:N7,MATCH(Q2,B3:B7,0),MATCH(Q3,C2:N2,0))

How the Formula Works


	The first MATCH function finds the row number in the range B3:B7 by looking for the sales person in the cell Q2. Abida is in row 2.




	The second MATCH function finds the column number in the range C2:N2 by looking at the month in cell Q3. June is in column 6




	The INDEX function is used to extract the value in the range C3:N7 by looking at the row and column numbers returned by the MATCH functions




	The answer is £2,076 because it is in row 2 and column 6 of the data set




Chapter 9: Sum all Characters in a Column

Objective

The objective is to sum all characters in a column.

There is a list of countries in the range B3:B10. The goal is to add all the characters in this range.

Functions Used


	SUMPRODUCT

	LEN



SUMPRODUCT Function Explained

The SUMPRODUCT function multiplies ranges together and returns the sum of the products. Using this function for its intended purpose is not the most useful but when you start using it to count and sum based on conditions means it becomes a very powerful function.

Syntax

=SUMPRODUCT(array1, [array2], [array3],…)

Arguments

array1 – The first range to multiply and then add

array2 – This is an optional argument. The second range to multiply and then add

array3 – This is an optional argument. The third range to multiply and then add

LEN Function Explained

See Chapter 6 – Sum all Digits in a String

The Formula to Sum all Characters in a Column

The formula to sum all the characters in the range B3:B10 is:

=SUMPRODUCT(LEN(B3:B10))

[image: ]


How the Formula Works


	LEN(B3:B10) section of the formula will return an array of 8 values. The result is {5;7;5;6;12;5;9;11}




	The SUMPRODUCT function will then simply sum all the values in the array



Points to Note


	This is an array formula but because the SUMPRODUCT function is used there is no need to press CTRL+SHIFT+ENTER like with other array formulas




	An alternative (and longer!) way to do this is to have a helper column in column C and in cell C3 write the formula =LEN(B3). Copy the formula down to cell C10 and then sum the results using the SUM function. As mentioned earlier in the book, this is one of the advantages of array formulas. You can take out intermediate formulas by using just one array formula.




Chapter 10: Sum Using SUMPRODUCT

Objective

Sum the CBM volumes moved by a shipping company by region and country using the SUMPRODUCT function

Function Used


	SUMPRODUCT



SUMPRODUCT Function Explained

See Chapter 9 – Sum all Characters in a Column

The Formula to Sum using the SUMPRODUCT Function

Cell G4 sums all the volumes together if the region in the range B3:B15 is South America. The formula is:

=SUMPRODUCT(--(B3:B15=G3),D3:D15)

How the Formula Works


	In the SUMPRODUCT formula in cell G4, B3:B15=G3 returns TRUE if the origin region is South America and FALSE if it isn’t in the range B3:B15. For South America it will return the array: {FALSE;FALSE;FALSE;FALSE;TRUE;FALSE;FALSE;



TRUE;FALSE;FALSE;FALSE;FALSE;FALSE}


	The -- dashes are used to change the TRUE values to a 1 and FALSE values to a 0. --(B3:B15=G3) returns the array {0;0;0;0;1;0;0;1;0;0;0;0;0}




	The SUMPRODUCT second argument (array2) D3:D15 returns {378;250;129;280;450;250;100;350;249;480;170;267;570}




	The formula is calculated as: =SUMPRODUCT({0;0;0;0;1;0;0;1;0;0;0;0;0}*{378;250;129;280;450;250;100;350;249;480;170;267;570})




	The formula then sums the two products of the arrays e.g. 0x378 + 0x250 + 0x129 + 0x280 + 1x450 and so on




	This returns 450 and 350




	The SUMPRODUCT function then adds the two values together to return 800




	If you want to sum multiple criteria’s then you have to include a plus sign (+) in the SUMPRODUCT's array1 argument followed by the second criteria you want to include



In cell G10 the formula =SUMPRODUCT((--(C3:C15=G8)+(C3:C15=G9)),D3:D15) is adding the volumes if the origin country is Spain and Brazil

Points to Note


	You can perform the same task using the SUMIF function




	
You don’t have to press CTRL+SHIFT+ENTER for this array formula as it is using the SUMPRODUCT function








Chapter 11: Calculate the MAX Values Based on Multiple Conditions

Objective

The objective is to calculate the max values based on a given number of conditions.

There is a data table which shows the region a Sales rep is in and the sales they have made. The goal is to calculate the max sales for Jane, Jim and Emma who are based in Europe. Notice they have more than one sales figure in column C and we need to extract the highest figure.

Functions Used


	MAX

	IF



MAX Function Explained

The MAX function returns the highest value in a range of cells. It only calculates numeric values. It ignores empty cells, TRUE and FALSE values and text.

Syntax

=MAX (number1, [number2], ...)

Arguments

number1 – The cell or range to extract the highest value

number2 – This is optional. The second cell or range where you want to extract the highest value

IF Function Explained

The IF function is a logical test function that returns a specified value if TRUE and another if the value is FALSE. You can nest multiple IF conditions and they are often used in conjunction with other logical functions such as the AND, OR functions.

Syntax

=IF (logical_test, [value_if_true], [value_if_false])

Arguments

logical_test – The value or cell you want to test if it is TRUE or FALSE

value_if_true – This is optional. The value to return if the logical_test is TRUE

value_if_false – This is optional. The value to return if the logical_test is FALSE

The Formula to Calculate the Max Value Based on Multiple Conditions

Cell G5 calculates the max sales value for Jane who is based in Europe. The formula is:

{=MAX(IF(($B$3:$B$13=$G$2)*($C$3:$C$13=F5),$D$3:$D$13))}

The formula is then copied down to cell G7

How the Formula Works


	If we use the formula in cell G5 as an example, ($B$3:$B$13=$G$2) checks which cell in the range $B$3:$B$13 is "Europe". This returns an array of TRUE and FALSE values, i.e. {TRUE;FALSE;FALSE;TRUE;TRUE;FALSE;TRUE;FALSE;FALSE;TRUE;TRUE}




	($C$3:$C$13=F5) checks which cell in the range $C$3:$C$13 is "Jane". This returns the array: {TRUE;FALSE;FALSE;TRUE;FALSE;FALSE;FALSE; FALSE;FALSE;FALSE;FALSE}




	The array converts the TRUE values to a 1 and the FALSE values to a 0




	$D$3:$D$13 returns the array: {585;895;452;622;856;954;256;478;125;658;568}




	The formula is calculated as: =MAX(IF(({1;0;0;1;0;0;0;0;0;0;0}*{585;895;452;622;856;954;256;478;125;658;568})




	The formula then sums the two products of the arrays e.g. 1x585 + 0x895 + 0x452 + 1x622 and so on




	This returns 585 and 622




	The MAX function then selects the higher of the two which is 622



Points to Note


	This must be entered as an array formula so you must press CTRL+SHIFT+ENTER otherwise the formula will not work




	
If you want the minimum value instead then change the MAX to a MIN function. If you want the average then change it to the AVERAGE function








Chapter 12: Extract the Middle Values of a Text String

Objective

The objective is to extract the middle section of a text string.

There is a list of product codes in the range B3:B6. The goal is to extract the middle section of the product codes.

Functions Used


	MID

	SEARCH



MID Function Explained

See Chapter 7 – Extract an Email Address from a Text String

SEARCH Function Explained

The SEARCH function locates the position of a character within a text string. It is very similar to the FIND function but the difference is that it is not case sensitive.

Syntax

=SEARCH(find_text, within_text, [start_num])

Arguments

find_text – The character or text to find

within_text – The text to search in

start_num – This is an optional argument. The starting location in the text to search

The Formula to Extract the Middle Section of a Text String

The formula in cell C3 is:

=MID(B3,SEARCH("-",B3)+1,SEARCH("-",B3,SEARCH("-",B3)+1)-SEARCH("-",B3)-1)

The formula is then copied down to cell C6

How the Formula Works


	Let’s take the formula in cell C3 as an example of how this works. The MID functions first argument is SEARCH("-",B3)+1. This returns 5, i.e. the position after the first "-".




	The MID functions second argument, SEARCH("-",B3,SEARCH("-",B3)+1)-SEARCH("-",B3)-1 finds the number of characters you want to extract, which is 5.




	The MID function therefore extracts the string starting from position 5 and extracts 5 characters from this position




	The MID function extracts 45RDF




Chapter 13: Useful Golden Nuggets about Functions

This chapter lists some useful information about Excel functions. For example, I have mentioned earlier in the book that when you create array formulas you have to enter CTRL+SHIFT+ENTER for most but not for all Excel functions. This chapter will explain which functions you will need to press CTRL+SHIFT+ENTER and which functions you don’t. This chapter will also provide you with some other great nuggets of information.

Functions that do not Require CTRL+SHIFT+ENTER for Array Formulas


	SUMPRODUCT

	AGGREGATE (no 14 to 19)

	LOOKUP

	INDEX



Function Arguments that can Handle Arrays Without CTRL+SHIFT+ENTER


	The array_1, array_2 etc. arguments in the SUMPRODUCT function

	The lookup_vector argument in the LOOKUP function

	The result_vector argument in the LOOKUP function

	The array argument in the INDEX function

	The array argument in the AGGREGATE function for functions 14 to 19



Function Arguments that can Contain Array Constants Without CTRL+SHIFT+ENTER

Remember array constants are items in curly brackets { } like this:

{1, 2, 3, 4, 5}


	The array argument in the SMALL function

	The array argument in the LARGE function

	The logical1 argument in the OR function

	The text argument in the LEN function

	The start_num argument in the MID function

	The number1 argument in the PRODUCT function

	The col_index_num argument in the VLOOKUP function

	The row_index_num argument in the HLOOKUP function

	The number1 argument in the SUM function



Function Arguments that can Contain Array Constants but still have to Press CTRL+SHIFT+ENTER


	The col_index_num argument in the VLOOKUP function



Array Functions

These are built in functions designed to deliver more than one item to more than one cell using CTRL+SHIFT+ENTER


	TRANSPOSE

	MODE.MULT (You do not need to press CTRL+SHIFT+ENTER)

	TREND

	FREQUENCY

	LINEST

	MMULT

	MINVERSE

	MUNIT



Function Arguments that can't Handle Arrays (Array Operations, Array Constants, and Arrays Created by Workbook References)


	The lookup_value argument in the VLOOKUP function

	The lookup_value argument in the HLOOKUP function

	The range argument in the SUMIF function

	The range argument in the COUNTIF function

	The range argument in the AVERAGEIF function

	The criteria_range argument in the SUMIFS function

	The criteria_range argument in the COUNTIFS function

	The criteria_range argument in the AVERAGEIFS function



Volatile Functions

Volatile functions recalculate every time Excel recalculates the spreadsheet. A spreadsheet can recalculate by entering new data, deleting or inserting a row or column, performing certain AutoFilter actions, double clicking a row or column divider, renaming a worksheet, changing a position of a worksheet or by forcing it by pressing F9 on your keyboard.


	CELL

	INDIRECT

	INFO

	NOW

	OFFSET

	RAND

	
TODAY








Additional Resources

For more information about Excel you can visit my website www.excelmasterconsultant.com
.

In my website you will find:


	Excel tutorials and blogs to expand your Excel knowledge




	Books you can buy that I read while I was learning Excel which has helped me to advance my Excel skills and which I recommend to you




	Excel applications you can buy which will make you more efficient and save you time




	Online Excel and VBA courses you can buy




	More information about my Excel consultancy services that I offer




	An online shop where you can buy books, applications and courses



Please take a visit and drop me a message. I would love to hear from you.


More Books by Excel Master Consultant

If you enjoyed reading this book then please look out for more Excel books by me.

If you want to learn more about some of my favourite and commonly used Excel formulas and functions then you can buy Excel Formulas and Functions: The Complete Excel Guide for Beginners.


This book includes:


	Tips on how to create more efficient formulas

	How the order of operator precedence in Excel Formulas work

	How to check formulas in your worksheets

	How to use the Excel Insert Function Formula Builder to make formulas easier to write

	How relative, absolute and mixed cell references are used to construct formulas

	How to create summing and counting formulas using Excel functions such as SUMIF, SUMIFS, COUNTIF, COUNTIFS, COUNTBLANK

	How to create logical formulas using the IF, OR, AND functions

	How to create lookup formulas using Excel functions such as VLOOKUP, HLOOKUP and combining the INDEX and MATCH functions together to create powerful lookups

	How to create text formulas by using Excel functions such as LEFT, RIGHT, MID, and CONCATENATE to manipulate text in your worksheets

	Great Excel hints and tips to help you become more efficient and save time



You can buy the book from Amazon by clicking on the link below.


https://www.amazon.com/Excel-Formulas-Functions-Complete-Beginners/dp/1699170932


Keep in Touch

Thank you for buying and reading this book. I hope you have enjoyed reading it as much as I have enjoyed writing it. I hope this book has been of great value and benefit to you and that you have learnt how to create some awesome formulas!

I love to hear from my readers so please get in touch with me. I am always looking to write more Excel books so please get in contact with me and let me know what area of Excel you would like help with and I will write a book about it. You can contact me on my website www.excelmasterconsultant.com
.

I hope to hear from you soon. In the meantime, happy Excel learning.

Harjit Suman

Excel Master Consultant

Leave a Review

What did you think of this book?

First of all, thank you for purchasing this book. I know you could have picked any number of Excel books to read, but you picked this book and for that I am extremely grateful.

I hope that it has improved your Excel skills and you are now able to master writing formulas in your worksheets. If so, it would be really nice if you could share this book with your friends and family by posting to Facebook
 and Twitter
. It would be nice if you can share what your favourite formulas were so your friends and family can try them out too!

If you enjoyed this book, I’d like to hear from you and hope that you could take some time to post a
 review on Amazon
. Your feedback and support will help me to greatly improve my writing craft for future projects and make this book even better.


You can leave a book review on Amazon here


I want you, the reader, to know that your review is very important to me and so, if you’d like to leave a review
, all you have to do is click here
 and away you go. I wish you all the best in your future success and happy Excel learning!


Your Free Bonus Gift!

As a small token of thanks for buying this book I would like to offer a FREE
 bonus gift to all my readers. I am offering a FREE
 online VBA macros course called How to Record Macros in Excel
.

In this FREE
 course you will learn:


	How to record a macro to eliminate manual repetitive Excel tasks

	How to execute a macro by:




	Using the Macro Dialog Box




	Using the Visual Basic Editor

	Clicking a button

	Clicking a shape



Once you have completed the course you will be able to:


	Automate Excel tasks easily

	Save valuable time

	Advance your Excel skills



You can register for this FREE
 online VBA course by clicking on the below link and entering your details now.


Register for the free VBA course

